
はじめに 

QlikViewにおける緑/白/グレーの色分け表示は非常にシンプルかつ直感的なコンセプトです。ユーザ
ーによる一連の選択操作によりアプリケーションのデータセットが絞り込まれ、グラフやレポートに即座に
反映されます。SET分析は、この機能を上書きし、オブジェクトやチャートレベルで選択操作をコントロ
ールすることが出来ます。SET分析により QlikViewアプリケーション開発者はチャートに表示すべきデ
ータを柔軟にコントロール出来るようになります。 

 

 SET分析を活用すると、例えば以下のようなことが出来るようになります。 

 前年対比など複数期間の比較 

 計算式における項目値の制限 

 現在の選択条件のを無視したり、いずれかの条件のみを含むような数式 

 累計計算や YTD(Year to Date)の計算 

 

SET分析構文は一見複雑に見えますが、コンセプトは極めてシンプルです。 

基本構文 

以下の例を使って、SET分析の構文がどのように構成されるかを説明します。 

 

– 現在の選択条件を含み、国(Country)=UKの売上(Sales)合計 

 

数式で使用する項目は Salesであり、それを合計するので Sum関数を使います。つまり以下の数式
から始めます。 

 

Sum(SALES) 
 

SET分析を表現する為に、括弧のすぐ後ろに {SET数式} として式を追加します。 

波括弧  {  の直後には、現在の選択を表す$記号または現在の選択を無視する 1などを記述します。
これを SET識別子と言います。この例では現在の選択が必要なので$を記述します。（赤字） 

 

Sum ({$}Sales) 
 

次に、SET数式で使われる項目を記述します。$記号の直後に <> 記号で括って記述します。（緑
字）この例では、国＝UKであるという条件のみが追記され、以下のようになります。 

 

Sum ({$ <Country ={“UK”} > } Sales) 
 

上記の数式を理解しやすいように複数行の構文に分割すると下記のようにも表現できます。 

 

Sum ( 
{$  
<Country ={“UK”} > 
 }  
Sales) 

SET数式の基本コンポーネント  

以下に SET数式における基本コンポーネントを説明します。 

SET 識別子 

レコード セットを表すために使用される定数を SET識別子 と言います。 

sum( {SET識別子} Sales) 

{$} ：現在の選択条件のレコードを表します。SET 数式 {$} は、SET 数式を提示していない 
 のと同じです。  

{1} ：現在の選択条件とは無関係にドキュメントのすべてのレコードセット全体を表します。 

{MyBookMark} ：BM01 や MyBookmark のように、ブックマーク ID もしくはブックマーク名で 
  指定された選択条件が使用されます。サーバーおよびドキュメント ブックマークのみ 
   SET 識別子として使用できます。  

$1 ： 一つ前の選択条件が使用されます。  

 

SET 修飾子 

SET修飾子は選択されたレコードセットを修正するために使用されます。 

sum( { $ <SET修飾子>} Sales) 

SET修飾子の構文 < field1 = selection1, field2=selection2, ... >  

例: 

sum( { $ < Year =  {2008} > } Sales)  2008年の売上  

sum( { $ < Year = Year -{2008} > } Sales) 選択年の売上 – 2008年の売上  

sum( { $ < Region = {“*”} > } Sales)  全地域の売上  

sum( { $ < Region = > } Sales)  地域の選択条件は無視  

SET 演算子 

+ ：和集合(Union)。この二項演算子は、2 つの SET オペランドの 
 いずれかに属するレコードを含む SET を返します。 

 

– ：排除(Exclusion)。2 つの内最初の SET オペランドにのみ属する 
 （もう一方には属さない）レコードを含む SET を返します。 
 また、単項演算子として使用する場合は、補集合を返します。 

 

* ：共通集合(Intersection)。2 つの SET オペランドの両方に属する 
 レコードを含む SET を返します。 

 

/ ： 排他的論理和または対象差(XOR)。2 つの SET オペランドの 
 どちらかに属していて、両方に属していないレコードを含む SET を 
 返します。  

 

SET 演算子を複数の QlikView テーブルからの項目を含む基本集計関数とともに使用するの
は、予測不可能な結果の原因となりますので、避けて下さい。  

例えば、Quantity と Price が別々のテーブルの項目の場合に、 
sum({$*BM01}Quantity*Price) という数式は避けて下さい。  

SET分析の例 

以下に一般的な SET分析の例を示します。 

 

通常の Salesの合計を表す数式： 
 

SUM( SALES) 
 
SET分析構文を使った Sales合計。$はすべての現在の選択条件を表す： 

 
SUM( {$} SALES) 

 
SET分析構文を使った Sales合計。1は現在の選択条件を無視することを表す： 

 
SUM( {1} SALES) 

 
現在の選択条件を含むすべての国の Sales合計： 

 
SUM( {$ <Country = {"*"} > } SALES) 

 

現在の選択条件を無視したすべての国の Sales合計： 
 
SUM( {1 <Country = {"*"} > } SALES) 

 

現在の選択条件を含む UKの Sales合計：  
 
SUM ({$ <Country ={“UK”} > } SALES) 

 
現在の選択条件を含む、UK以外のすべての国の Sales合計： 

 
SUM ({$ <Country -={“UK”} > } SALES) 

動的なレコードセット 

ある時点での売上を表現するには SET分析が便利です。例えば 2010年の売上を表すには以下の
数式を使います。 

Sum( 

 {$ 

  <Year = {2010}> 

   } 

     Sales) 

 

ユーザーが選択した年度に応じてその 1年前を表現するには、どうしたら良いでしょう？ 

 

前年を表す関数、 Max(Year) – 1を使います。この例では、ユーザーが 2010年を選択すれば関数
の結果は 2009 となります。複数年を選択したり、どの年度も選択しなかった場合は最新の年に基づ
いて計算されます。 

 

この関数を SET修飾子として使います。 

 

$(=Function())   

 

最終的には以下のような数式となります。 

 

Sum( 

 {$ 

  <Year = {$(=Max(Year) - 1)}> 

   }  

   Sales) 

 

このアプローチは四半期や月、日など他の項目についても同じように使えます。しかし、例えば月につい
ていえばユーザーが 1月を選択した場合は Max(Month)-1の結果がゼロとなるため、期待する結果と
はならないことに注意して下さい。 

選択条件への値の追加/除外 

次の演算子を使ってユーザー操作による選択値を修正出来ます。 

 

=   項目に対する選択値を指定します。 

+= 選択された項目値と指定された値の和集合を指定します。 

 

Sum( 

 {$ 

  <Year += {2009, 2010}> 

   }  

    Sales) 

 ……上記の数式はユーザーが選択した年に加え、2009年と 2010年の売上合計を返します。 

 

 

-= ユーザーが選択した値から指定した値を除外した値を指定します。 

Sum( 

 {$ 

  <Product -= {‘Product X’}> 

   } 

     Sales)  

…… 上記の数式はユーザーが選択した Productから Product Xを除いた売上を返します。 

 

 

*= ユーザーが選択した値と数式で指定された値の共通部分を指定します。 

Sum( 

 {$ 

  <Product *= {‘Product X’}> 

   } 

     Sales)  

……上記の数式は現在の選択のうち Product の現在の選択値と Product Xの共通部分における
売上合計を返します。 

 

  



 

Page | 10  
 

/= 排他的論理和(XOR)、つまり共通部分とは逆にどちらかにしか存在しない値を指定します。 

Sum( 

 {$ 

  <ProductNumber /= {“33*”}> 

   } 

     Sales) 

 …….上記の数式は現在の選択のうち ProductNumberの現在の選択値と 33で始まる
ProductNumberのどちらかにのみ当てはまる売上合計を返します。 

その他の例 
 

– Customer Countryおよび Policy Statusの選択を無視した売上合計： 

 SUM( {$<[Customer Country] = , [Policy Status] = >} Sales )  
 
 
– Policy Created Yearが変数 vMaxCreatedYearの値と同じ年の売上合計： 

 SUM( {$<[Policy Created Year] = {$(#vMaxCreatedYear)}>} Sales ) 
 
 
– Policy Created Yearが変数 v2002の値以降の年の売上合計： 

 SUM( {$<[Policy Created Year] = {">=$(#v2002)"}>} Sales ) 
 
 
– Policy Created Yearが変数 v2002の値より前の年の売上合計： 

 SUM( {$<[Policy Created Year] = {"<$(#v2002)"}>} Sales ) 
 
 
– Policy Created Yearが変数 v2002の値から変数 v2004の値までの年の売上合計： 

 SUM( {$<[Policy Created Year] = {">=$(#v2002) <=$(#v2004)"} >} Sales ) 
 
 
–  Productが Life Planではない製品の売上合計： 

 SUM( {$<Product -= {"Life Plan"} >}  Sales  ) 
 
 
– Policy Created Yearが 2000年から 2003年までの売上合計： 

 SUM( {$<[Policy Created Year]  =  {“>1999<2004”}>}  Sales ) 
 
 
– Policy Created Yearが 200X年の売上合計： 

 SUM( {$<[Policy Created Year]  =  {“200*”}>} Sales ) 
 
 
– 名前に Planが含まれる製品の売上合計： 

 SUM(  {$<Product  = {“*Plan*”}>}  Sales ) 
 
 
– 現在の選択において、選択された製品に加え下記の 2製品（Life Plan、Fleet Secure）を
加えた売上合計： 

 SUM( {$<Product += {"Life Plan", "Fleet Secure"} >} Sales )  

– 現在の選択において、選択された年に加え 20XX年と 1997年を加えるが 2000年のみを除
いた年の売上合計：   

 SUM( {$<[Policy Created Year]  += {“20*”,1997} – {2000} >} Sales ) 
 

 

– 現在の選択において、選択された製品と Fleet Secure という製品の共通部分の売上合計： 

 SUM( {$<Product *= {"Fleet Secure"} >} Sales )  
 

 

– 現在の選択年の前年の売上合計。$記号展開が前年の計算に使われています。： 

 SUM( {$<Year = {$(#=Only(Year)-1)}>} Sales )  
 

 

– 現在の選択において、Planまたは Secure という名前の製品を除く売上合計： 

 SUM({$-1<Product = {"*Plan*", "*Secure*"}>} Sales )  
 

 

– 現在の選択において、2007年の売上が 1,000,000以上の Customerのみを含む売上合
計： 

 SUM( 
  {$ 
 <Customer = {“=Sum({1<Year = {2007}>} Sales ) > 1000000”}> 
 }  
 Sales ) 

